

Nombre

DNI

Grupo

INSTRUCCIONES

- Todas las preguntas del test se han de contestar en la tabla a pie de página de esta primera hoja en MAYÚSCULAS. Las respuestas fuera de la tabla no serán tenidas en cuenta.
- Todas las respuestas se realizarán en las hojas grapadas del examen en las zonas reservadas para ello.
- Sólo se utilizará una carilla de la hoja para contestar las cuestiones.
- Se pueden usar los reversos como hojas a sucio.
- No se pueden desgrapar las hojas ni añadir nuevas hojas.
- Sólo hay una respuesta válida en cada pregunta de test.
- Cada pregunta del test puntua 0,4 puntos en caso de acierto. Las incorrectas restan el valor de media pregunta (0,2). Las no contestadas no puntúan.
- La duración del examen será de 2 horas.

TEST

1 Sabiendo que la información en el ordenador se almacena finalmente como unos y ceros, ¿cuál de las siguientes afirmaciones es cierta?

- A Sólo se pueden almacenar números pero no caracteres.
- B Se pueden almacenar tanto números como caracteres codificados.
- C La información básica no consta sólo de unos y ceros, sino que tiene un alfabeto mucho más amplio.
- D Ninguna de las anteriores

2 ¿Diferencia entre un MODEM y una tarjeta de red?

- A El MÓDEM es siempre un dispositivo externo y la tarjeta de red interno
- B A diferencia de la tarjeta, el MÓDEM usa la red telefónica básica por medio de señales acústicas.
- C El MÓDEM puede comunicarse a velocidades mayores que haría una tarjeta de red.
- D El MÓDEM comunica ordenadores entre si y la tarjeta de red sólo conecta ordenadores con HUBs

3 ¿Para qué se usa la expresión `ReDim Preserve`?

- A Para declarar matrices estáticas
- B Para declarar matrices dinámicas
- C Para cambiar las dimensiones de una matriz dinámica sin que se pierdan los contenidos
- D Para poner a cero una matriz estática o liberar la memoria asociada si es dinámica

4 ¿Qué valor muestra en pantalla el siguiente programa?

- A 36
- B 24
- C 0

```
Sub main()
  Dim A(1 To 3, 1 To 3) As Integer
  Dim i As Integer
  For i = 1 To 3
    For j = i To 3
      A(i, j) = i + j
      suma = suma + A(i, j)
    Next j
  Next i
  MsgBox (suma)
End Sub
```

- D Hay un error sintáctico y el programa no se ejecuta.

PREGUNTA	1	2	3	4	5	6	7	8	9	10
RESPUESTA	B	B	C	B	B	C	A	A	B	D

5 ¿Qué valor muestra en pantalla el siguiente programa?

- A 6
- B 0
- C Hay un error sintáctico y el programa no se ejecuta.
- D Ninguna de las anteriores

```
Sub main()  
  Dim v() As Integer  
  ReDim v(1 To 3) As Integer  
  For i = 1 To 3  
    v(i) = i  
  Next i  
  ReDim v(1 To 4)  
  suma = v(1)+v(2)+v(3)+v(4)  
  MsgBox (suma)  
End Sub
```

6 La expresión en Visual Basic `contador = contador+i`

- A Incrementa la variable `contador` en 1 cada vez que se ejecuta.
- B Es incorrecta ya que utiliza como nombre de variable una palabra reservada de Visual Basic
- C Incrementa la variable `contador` en el valor de la variable `i` cada vez que se ejecuta.
- D Ninguna de las anteriores contestaciones es correcta

7 ¿Qué significa la expresión `Dim s(14) As String ?`

- A Declara un vector de `Strings` donde el índice puede tomar un valor máximo de 14.
- B Declara una `String` de tamaño fijo, concretamente de 14 caracteres.
- C Esa sentencia no tiene sentido ya que contiene un error sintáctico.
- D Ninguna de las anteriores.

8 ¿Qué valor muestra en pantalla el siguiente programa?

- A 0
- B 3
- C 6
- D Se produce un error al ejecutar la función.

```
Function f (n As Integer) As Integer  
  Dim i As Integer, suma As Integer  
  For i = 1 To n  
    suma = suma + i  
  Next i  
End Function  
  
Sub Main ()  
  MsgBox (f(3))  
End Sub
```

9 ¿Qué valor muestra en pantalla el siguiente programa?

- A 0
- B 3
- C 6
- D Se produce un error porque el modificador `ByVal` no se puede usar de ese modo.

```
Sub doble(ByVal n As Integer)  
  n = n * 2  
End Sub  
  
Sub main()  
  Dim i As Integer  
  i = 3  
  Call doble(i)  
  MsgBox (i)  
End Sub
```

10 ¿Qué función realiza el procedimiento `AddItem` en el control `ComboBox`?

- A Ese procedimiento existe en el `Listbox`, pero no en el `ComboBox`.
 - B Añade un valor nuevo al vector `ItemData` del `ComboBox`
 - C Selecciona un elemento del vector `List` del `ComboBox`.
 - D Ninguna de las anteriores.
-

PROBLEMA 1 DE VISUAL BASIC (4 puntos)

Se han de resolver los siguientes apartados sin escribir ningún procedimiento *Sub Main()*.

1. Un fichero de **acceso aleatorio** contiene los nombres, códigos postales (enteros) y número de habitantes de un conjunto de ciudades. Definir un tipo de dato para almacenar la información de una ciudad de modo que permita acceder a la información de ese archivo. (0,5 puntos).
2. Escribir un **procedimiento** que guarde en un **fichero de texto** el nombre, el código postal y el número de habitantes de la ciudad del fichero de acceso aleatorio con mayor población. Los nombres de ambos ficheros se recibirán como argumentos. (2 puntos).
3. Escribir una **función** que tenga como argumentos el código postal de la ciudad y el nombre de un fichero y que devuelva un valor *booleano* que indique si una ciudad concreta se encuentra en el fichero o no. (1,5 puntos).

SOLUCIÓN

```
'APARTADO 1
Type Ciudad
    nombre As String * 20
    codigo_postal As Integer
    habitantes As Long
End Type

' APARTADO 2
Sub guarda(nft As String, nfa As String)
    Dim numFichAlea As Integer, numFichTxt As Integer
    Dim c As Ciudad, n As Integer, i As Integer, cmax As Ciudad

    numFichAlea = FreeFile
    Open nfa For Random As #numFichAlea Len = Len(c)
    n = LOF(numFichAlea) / Len(c)
    For i = 1 To n
        Get #numFichAlea, i, c
        If (c.habitantes >= cmax.habitantes) Then cmax = c
    Next i
    Close #numFichAlea

    numFichTxt = FreeFile
    Open nft For Output As #numFichTxt
    With cmax
        Write #numFichTxt, .nombre, .codigo_postal, .habitantes
    End With
    Close #numFichTxt
End Sub

' APARTADO 3
Function Busca(nomFich As String, codigo As Integer) As Boolean
    Dim numF As Integer, c As Ciudad, n As Integer, i As Integer

    numF = FreeFile
    Open nomFich For Random As #numF Len = Len(c)
    n = LOF(numF) / Len(c)
    Busca = False
    For i = 1 To n
        Get #numF, i, c
        If (c.codigo_postal = codigo) Then
            Busca = True
            Exit For 'podría omitirse esta sentencia
        End If
    Next i
    Close #numF
End Function
```

PROBLEMA 2 DE VISUAL BASIC (2 puntos)

Escribir un programa que realice lo siguiente:

- Solicitar al usuario un valor n que sea múltiplo de 3 y esté entre 6 y 21, ambos incluidos. Repetir la petición hasta que el número introducido sea válido.
- Siendo n el valor entero indicado en el apartado anterior, crear una **matriz** de n filas y $n-1$ columnas que permita almacenar registros como los del apartado 1 del problema anterior.
- Pedir al usuario los datos de cada celda de la matriz.
- Mostrar la suma total de los habitantes de la ciudad menos poblada de cada columna.

SOLUCIÓN

```
Option Explicit
```

```
Type Ciudad
```

```
    nombre As String * 20 ' valdría también sin '* 20'  
    codigo_postal As Integer  
    habitantes As Long
```

```
End Type
```

```
Sub main()
```

```
    Dim n As Integer           ' APARTADO A  
    Dim matriz() As Ciudad    ' APARTADO B  
    Dim i As Integer, j As Integer ' APARTADO C  
    Dim suma As Long, minColumna As Long ' APARTADO D
```

```
    ' APARTADO A
```

```
    While (n Mod 3 <> 0 Or n < 6 Or n > 21)
```

```
        n = InputBox("Introduzca un entero múltiplo de 3, entre 6 y 21")
```

```
    Wend
```

```
    ' APARTADO B
```

```
    ReDim matriz(1 To n, 1 To n - 1)
```

```
    ' APARTADO C
```

```
    For i = 1 To n
```

```
        For j = 1 To n - 1
```

```
            With matriz(i, j)
```

```
                .nombre = InputBox("¿Nombre de la ciudad?")
```

```
                .codigo_postal = InputBox("¿Código postal?")
```

```
                .habitantes = InputBox("¿Número de habitantes?")
```

```
            End With
```

```
        Next j
```

```
    Next i
```

```
    ' APARTADO D
```

```
    For j = 1 To n - 1
```

```
        minColumna = matriz(1, j).habitantes
```

```
        For i = 2 To n
```

```
            If (matriz(i, j).habitantes < minColumna) Then
```

```
                minColumna = matriz(i, j).habitantes
```

```
            End If
```

```
        Next i
```

```
        suma = suma + minColumna
```

```
    Next j
```

```
    MsgBox ("Suma de los menores de cada columna: " & suma)
```

```
End Sub
```